Lab-Report

ECAD

Analogue Amplitude Modulation and Demodulation with (P)Spice



Name: Dirk Becker
Course: BEng 2
Group: A
Student No.: 9801351
Date: 30/10/1998

UNIVERSITY of EAST LONDON

Contents

I. Behavioural model of amplitude modulation and demodulation

1) Introduction

Amplitude modulation is used since the first days of the 20th century mainly for transmitting voice and signals through the conventional broadcast band like the long-, medium- and short-wave bands because of its easy and cheap way of realisation.
Besides the consumption of bandwidth in comparison to usual FM is relatively small and the receivers could be made up very simple.

2) Standard AM modulator/demodulator model



Figure 1 Standard AM modulation/demodulation block diagram

A behavioural model of a standard amplitude modulation/demodulation consists of the modulator and the demodulator section (Figure 1).
In the modulator section first the modulating signal (node 1) and a DC-offset (node 2) are added and then mixed with the carrier signal (node 4) by means of an analogue multiplier. Thus the modulating signal is shifted to the frequency of the carrier signal and can be taken off at the output of the analogue multiplier (node 5).
The information of the modulation signal is contained in the two side-bands with ($\omega_c+\omega_{mod}$) and (ω_c- ω_{mod}).
The carrier ω_c contains no useful information.

Hence Standard AM modulation can be expressed with the following equation:

$$V_m = A\sin(\omega_c t)(1 + m\cos(\omega_{mod} t))$$

$$\text{where } m = \left(\frac{\text{peak modulating signal}}{\text{peak carrier signal}}\right) \text{and } A = \text{amplitude of carrier signal}$$

At this stage only ideal components are used for demonstrating AM modulation and demodulation.

Mathematical justification of standard amplitude modulation:

$$V_m = A\sin(\omega_c t)(1 + m\cos(\omega_{mod} t))$$

$$V_m = A\left(-\frac{1}{2j}\right)\left(e^{j\omega_c t} - e^{-j\omega_c t}\right)\left(1 + m\frac{1}{2}\left(e^{j\omega_{mod} t} + e^{-j\omega_{mod} t}\right)\right)$$

$$V_m = A\left(-\frac{1}{2j}\right)\left(e^{j\omega_c t} - e^{-j\omega_c t}\right)\left(1 + m\frac{1}{2}\left(e^{j\omega_{mod} t} + e^{-j\omega_{mod} t}\right)\right)$$

$$V_m = -\frac{A}{2j}\left(e^{j\omega_c t} + e^{-j\omega_c t}\right) - \frac{Am}{4j}\left(e^{j\omega_{mod} t} + e^{-j\omega_{mod} t}\right)\left(e^{j\omega_c t} - e^{-j\omega_c t}\right)$$

$$V_m = -\frac{A}{2j}\left(e^{j\omega_c t} + e^{-j\omega_c t}\right) - \frac{Am}{4j}\left(e^{j\omega_{mod} t}e^{j\omega_c t} - e^{j\omega_{mod} t}e^{-j\omega_c t} + e^{-j\omega_{mod} t}e^{j\omega_c t} - e^{-j\omega_{mod} t}e^{-j\omega_c t}\right)$$

$$V_m = -\frac{A}{2j}\left(e^{j\omega_c t} + e^{-j\omega_c t}\right) - \frac{Am}{4j}\left(e^{j(\omega_{mod}+\omega_c)t} - e^{j(\omega_{mod}-\omega_c)t} + e^{j(\omega_c-\omega_{mod})t} - e^{-j(\omega_{mod}+\omega_c)t}\right)$$

$$V_m = -\frac{A}{2j}\left(e^{j\omega_c t} + e^{-j\omega_c t}\right) - \frac{Am}{4j}\left(e^{j(\omega_{mod}+\omega_c)t} - e^{-j(\omega_{mod}+\omega_c)t} + e^{j(\omega_{mod}-\omega_c)t} - e^{-j(\omega_{mod}-\omega_c)t}\right)$$

$$V_m = A\sin(\omega_c t) - \frac{Am}{4j}\left[-2j\sin((\omega_{mod} + \omega_c)t) - 2j\sin((\omega_{mod} - \omega_c)t)\right]$$

$$V_m = A\sin(\omega_c t) + \frac{Am}{2}\left[\sin((\omega_{mod} + \omega_c)t) + \sin((\omega_{mod} - \omega_c)t)\right]$$

Thus the following frequency components appear:

$$A = 1,\ m = 1$$

$$f_c = 101\text{kHz}$$

$$f_{mod} = 1\text{kHz}$$

$$V_m = \sin(2\times 101\text{kHz}\times\pi\times t) + \frac{1}{2}\begin{pmatrix}\sin(2\times 101\text{kHz}\times\pi\times t - 2\times 1\text{kHz}\times\pi\times t) + \\ \sin(2\times 101\text{kHz}\times\pi\times t - 2\times 1\text{kHz}\times\pi\times t)\end{pmatrix}$$

occurring frequencies:

1.) $f_{mod} = 101\text{kHz}$ at amplitude A

2.) $f_{mod} + f_c = 102\text{kHz}$ at amplitude $\frac{1}{2}A$

3.) $f_{mod} + f_c = 102\text{kHz}$ at amplitude $\frac{1}{2}A$

The (P)Spice listing, a transient plots at nodes 5, 6, 1, 7 and a FFT-analysis can be found on the following pages.

```
Standard Amplitude Modulation
*
*modulating signal at nodes 1 and GND (0)
*
vmod 1 0 sin(0 0.5 1k 0 0 90)
*
*dc input voltage at node 2
*
vdc 2 0 dc 1
*
*carrier signal at node 4
*
vcarr 4 0 sin(0 1 101k)
*
*summer with output towards node 3
*
esum 3 0 value {v(1)+v(2)}
*
*multiplier with output towards node 5
*
emult 5 0 value {v(3)*v(4)}
*
*1/2 wave rectifier model
*
erectifier 6 0 value {v(5)*(exp(10*v(5))/exp(abs(10*v(5))))}
*
*2nd order butterworth filter to remoove high freuqencies
*parameter definitions for H(s)=a1_/(a1 + a2*s + s*s)
*
.param q=0.707
.param omega=1
.param fc=1k
.param pi=3.14
.param pisquare ={3.14*3.14}
.param a1 ={4*pisquare*omega*omega*fc*fc}
.param a2 ={2*pi*omega*fc/q}
*
*realisation of H(s) with node 7 as output and node 6 as input
*
elow 7 0 LAPLACE {V(6)} = {a1/(a1+a2*s+s*s)}
*
*transient analysis
*
.tran 0.1m 20m 0 .1m
.probe
.end
```

3) Balanced amplitude modulation/demodulation System



Figure 2 Balanced AM modulation/demodulation block diagram

Using balanced amplitude modulation the transmission of the non information covering carrier signal is suppressed.
Therefore the DC-offset of figure 1 is let out, the carrier (node 2) and the modulating signal (node 1) of figure 2 are directly multiplied together.
At the output of the analogue multiplier appear only the two side-bands (ω_c+ω_{mod}) and (ω_c-ω_{mod}).
Balanced amplitude modulation needs a special kind of demodulation via multipliers to get back the original modulation signal content. By rectifier demodulation the output is not equivalent to the input signal.
The demodulation has to be done by means of an multiplier too (see next part).

Balanced AM can be described as:

$$V_m = A\sin(\omega_c t) m\cos(\omega_{mod} t)$$

Mathematical justification for the balanced amplitude modulation:

$$V_m = A\sin(\omega_c t) m\cos(\omega_{mod} t)$$

$$V_m = A\left(-\frac{1}{2j}\right)\left(e^{jw_c t} - e^{jw_{mod} t}\right)\frac{1}{2}\left(e^{jw_{mod} t} + e^{-jw_{mod} t}\right)$$

$$V_m = -\frac{Am}{4j}\left(e^{j(w_c+\omega_{mod})t} - e^{-j(w_c+\omega_{mod})t} + e^{j(w_c-\omega_{mod})t} e^{-j(w_c-\omega_{mod})t}\right)$$

$$V_m = -\frac{Am}{4j}\left(-\frac{1}{2j}\sin(\omega_c+\omega_{mod}) - \frac{1}{2j}\sin(\omega_c-\omega_{mod})\right)$$

$$V_m = -\frac{Am}{2j}\left(-j\sin(\omega_c+\omega_{mod}) - j\sin(\omega_c-\omega_{mod})\right)$$

$$V_m = \frac{A\times m}{2}\left(\sin(\omega_c+\omega_{mod}) + \sin(\omega_c-\omega_{mod})\right)$$

With the frequencies of the following (P)Spice simulation:

$$A = 1, m = 1$$
$$f_c = 101kHz$$
$$f_{mod} = 1kHz$$
$$V_m = \frac{1}{2}\begin{pmatrix} \sin(2 \times 101kHz \times \pi \times t - 2 \times 1kHz \times \pi \times t) + \\ \sin(2 \times 101kHz \times \pi \times t - 2 \times 1kHz \times \pi \times t) \end{pmatrix}$$

occurring frequencies:

1.) $f_{mod} + f_c = 100kHz$ at amplitude $\frac{1}{2}A$

2.) $f_{mod} + f_c = 102kHz$ at amplitude $\frac{1}{2}A$

The (P)Spice listing, a transient plot and a FFT-analysis at node 5 can be found on the following pages.

```
Balanced Amplitude Mdulation with standard demodulation
*
*modulating signal
*
vmod 1 0 sin(0 0.5 1k 0 0 90)
*
*carrier siggnal
*
vcarr 2 0 sin(0 1 101k)
*
*multiplier
*
emult 3 0 value {v(1)*v(2)}
*
*1/2 wave rectifier model
*
erectifier 4 0 value {v(3)*(exp(10*v(3))/exp(abs(10*v(3))))}
*
*2nd order butterworth filter to remoove high freuqencies
*parameter definitions for H(s)=a1_/(a11 + a2*s + s*s)
*
.param q=0.707
.param omega=1
.param fc=1k
.param pi=3.14
.param pisquare ={3.14*3.14}
.param a1 ={4*pisquare*omega*omega*fc*fc}
.param a2 ={2*pi*omega*fc/q}
*
*realisation of H(s) with node 7 as output and node 6 as input
*
elow 5 0 LAPLACE {V(4)} = {a1/(a1+a2*s+s*s)}
*
*transient analysis
*
.tran 0.1m 10m 0 .1m
.probe
.end
```

4) Model of multiplier demodulation



Figure 3 AM modulation/demodulation block diagram with two multipliers

The same way the modulation can be done with a multiplier the demodulation can be done too. The AM modulated signal must only be multiplied again by the carrier frequency. At the output of this 2nd multiplier are ($2\omega_c+\omega_{mod}$) , ($2\omega_c-\omega_{mod}$) and ω_c present, where the unnecessary frequencies are suppressed by means of a lowpass filter.

Mathematical justification for the multiplier demodulation:

$$V_{dem} = \left\{ A\sin(\omega_c t) + \left[\frac{Am}{2} \left(\sin((\omega_c + \omega_{mod})t) + \sin((\omega_c + \omega_{mod})t) \right) \right] \right\} \times \sin(\omega_c t)$$

$$V_{dem} = \left[-\frac{A}{2j}\left(e^{j\omega t} - e^{-j\omega t}\right) - \frac{Am}{4j}\left(e^{j(\omega_c+\omega_{mod})t} - e^{-j(\omega_c+\omega_{mod})t} + e^{j(\omega_c-\omega_{mod})t} - e^{-j(\omega_c-\omega_{mod})t}\right) \right] \times$$
$$\times \left(-\frac{1}{2j}\left(e^{j\omega_c t} - e^{-j\omega_c t}\right) \right)$$

$$V_{dem} = -\frac{A}{4}\left(e^{j2\omega_c t} - 2 + e^{-j2\omega_c t}\right) - \frac{Am}{8}\begin{pmatrix} e^{j(2\omega_c+\omega_{mod})t} - e^{-j(2\omega_c+\omega_{mod})t} - 2e^{j\omega_{mod}t} - 2e^{-j\omega_{mod}t} + \\ + e^{j(2\omega_c-\omega_{mod})} + e^{-j(2\omega_c-\omega_{mod})} \end{pmatrix}$$

$$V_{dem} = -\frac{A}{4}\left(2\cos(\omega_c t) - 2\right) - \frac{Am}{8}\begin{bmatrix} 2\cos((2\omega_c + \omega_{mod})t) - 4\cos(\omega_{mod}t) + \\ + 2\cos((2\omega_c - \omega_{mod})t) \end{bmatrix}$$

$$V_{dem} = -\frac{A}{2}\left(\cos(\omega_c t) - 1\right) - \frac{Am}{4}\begin{bmatrix} \cos((2\omega_c + \omega_{mod})t) - 2\cos(\omega_{mod}t) + \\ + \cos((2\omega_c - \omega_{mod})t) \end{bmatrix}$$

5) Standard Diode - Demodulation

In the first listing only a simple Diode was used for demodulation. This is only correct possible at Standard AM modulation. Trying to demodulate balanced AM with a diode demodulator will fail.
The diode demodulation in the listing was done by means of the simulation of the exponential characteristics of the diode. The negative parts of the signal are cut off (rectified) and the lowpass filter filters the remaining carrier frequencies out.

$$A = 1,\ m = 1$$
$$f_c = 10\text{kHz}$$
$$f_{mod} = 1\text{kHz}$$
$$V_{dem} = -\frac{A}{2}\left[\cos(2\omega_c t) - 1\right] - \frac{Am}{4}\begin{bmatrix} \cos((2\omega_c + \omega_{mod})t) - \\ -2\cos(\omega_{mod} t) + \\ +\cos((2\omega_c - \omega_{mod})t) \end{bmatrix}$$

occurring frequencies at node 6 :

1.) $2 \times f_c = 20\text{kHz}$ at amplitude $\frac{1}{2}A$

2.) DC at level $\frac{1}{2}A$

3.) $2 \times f_c + f_{mod} = 21\text{kHz}$ at amplitude $\frac{1}{4}A$

4.) $2 \times f_c - f_{mod} = 19\text{kHz}$ at amplitude $\frac{1}{4}A$

5.) $f_{mod} = 1\text{kHz}$ at amplitude $\frac{1}{2}A$

where 2.) and 5.) the original modulating signal at half amplitude represent (DC because of standard amplitude modulation)

The (P)Spice listing, a transient plot and a FFT-analysis at node 5 can be found on the following pages.

```
Standard Amplitude Mdulation with multiplier demodulation
*
*modulating signal at nodes 1 and GND (0)
*
vmod 1 0 sin(0 0.5 1k 0 0 90)
*
*dc input voltage at node 2
*
vdc 2 0 dc 1
*
*carrier signal at node 4
*
vcarr 4 0 sin(0 1 10k)
*
*summer with output towards node 3
*
esum 3 0 value {v(1)+v(2)}
*
*multiplier with output towards node 5
*
emult 5 0 value {v(3)*v(4)}
*
*multiplier demodulation
*
emult_2 6 0 value {v(5)*v(4)}
*
*2nd order butterworth filter to remoove high freuqencies
*parameter definitions for H(s)=a1_/(a1 + a2*s + s*s)
*
.param q=0.707
.param omega=1
.param fc=1k
.param pi=3.14
.param pisquare ={3.14*3.14}
.param a1 ={4*pisquare*omega*omega*fc*fc}
.param a2 ={2*pi*omega*fc/q}
*
*realisation of H(s) with node 7 as output and node 6 as input
*
elow 7 0 LAPLACE {V(6)} = {a1/(a1+a2*s+s*s)}
*
*transient analysis
*
.tran 0.1m 20m 0 .1m
.probe
.end
```

II. Modulation and demodulation using “real” parts

1) Modulation via Gilbert Cell

The Gilbert Cell is an discrete analogue multiplier. It can replace the ideal multipliers in chapter 3 and 4, the multiplier modulator and the multiplier demodulator.
At a time only one of the two multipliers is replaced, because otherwise special coupling facilities have to be connected between the two Gilbert Cells in order to prevent interference between the Gilbert Cells.



Figure 4 Modulation using Gilbert Cell

The modulating Gilbert Cell:



Figure 5 Schematics of the modulating Gilbert Cell

V5 is the input of the carrier signal and V4 the input of modulating signal. The inputs are coupled with voltage controlled voltage sources (nodes 51, 52 and nodes 11, 12) to separate the Gilbert Cell from the rest of the circuit.
The signals are then multiplied together and the output is done by subtraction of node 110 and node 120 via voltage controlled voltage source too (Esubtract in the Spice listing).
The following listing provides the replacement of the first multiplier via sub circuit Gilbert Cell as well as a Spice transient and FFT analysis of the concerned nodes.

The occurring frequencies are of course the same, than the ones with the ideal multiplier.

```
Standard Amplitude Modulation with multiplier demodulation and Gilbert Cell
modulation
*
*Subcircuit of Gilbert Cell
*
.subckt gilbert_cell 130 150 170 180 110 120
.model Q2 NPN (bf=100 is=1e-16)
      R1 110 100  5k
      R2 120 100  5k
      R3 200 190  15k
      R4 210 200  15k
      Q1 140 170 190 Q2
      Q2 160 180 210 Q2
      Q3 110 130 140 Q2
      Q4 120 150 140 Q2
      Q5 110 150 160 Q2
      Q6 120 130 160 Q2
      I1 200 220 DC 200uA
      V2 220 0 -10V
      V3 100 0 10V
.ends gilbert_cell
*
*modulating signal at nodes 1 and GND (0)
*
vmod 1 0 sin(0 0.5 1k 0 0 90)
*
*dc input voltage at node 2
*
vdc 2 0 dc 1
*
*carrier signal at node 4
*
vcarr 4 0 sin(0 1 101k)
*
*summer with output towards node 3
*
esum 3 0 value {v(1)+v(2)}
*
*first mutiplier (Gilbert Cell)
*
ecouple 51 52 value {v(4)}
ecoupl2 11 12 value {v(3)}
xmult 52 51 11 12 30 40 gilbert_cell
esubtractor 5 0 value {v(30)-v(40)}
*
*multiplier demodulation
*
emult_2 6 0 value {v(5)*v(4)}
*
*2nd order butterworth filter to remoove high freuqencies
*parameter definitions for H(s)=a1_/(a1 + a2*s + s*s)
*
.param q=0.707
.param omega=1
.param fc=1k
.param pi=3.14
.param pisquare ={3.14*3.14}
.param a1 ={4*pisquare*omega*omega*fc*fc}
.param a2 ={2*pi*omega*fc/q}
*
*realisation of H(s) with node 7 as output and node 6 as input
*
elow 7 0 LAPLACE {V(6)} = {a1/(a1+a2*s+s*s)}
*
*transient analysis
*
.tran 0.1m 2m 0 0.1m
.probe
.end
```

2) Demodulation via Gilbert Cell

Now the Gilbert Cell replaces the second multiplier. But this Gilbert Cell is only working well at small signals. It is less complex than the other one: The negative source and the emitter resistors in the lower part are missing.



Figure 6 Deodulating with Gilbert Celll - Block diagram

The demodulating Gilbert Cell:

V4 is the input for the carrier and V3 is the input for the modulated carrier. The Gilbert Cell multiplies both signals and the output can be taken of differential from nod 23 and node 24. In the listing it is done by means of the esubtractor function, a voltage controlled voltage source driven by the subtraction of the two upper collector potentials.



Figure 7 Standard AM modulation/demodulation block diagram

Although the occurring frequencies are the same than in the ideal multiplier curcuit a transient and a FFT analysis can be following on the following pages.

```
Standard Amplitude Modulation with multiplier modulation and
Gilbert Cell demodulation
*
*subcircuit Gilbert Cell
*
.subckt gilbert_demod 50 10 40 23 24
.model QStd NPN (BF=100 IS=1E-16)
     R1 23 100  5k
     R2 24 100  5k
     Q1 23 50 22 QStd
     Q2 24 0  22 QStd
     Q3 23 0  25 QStd
     Q4 24 50 25 QStd
     Q5 22 10 12 QStd
     Q6 25 40 12 QStd
     VB 40  0 -2V
     V1 100 0 10V
     I1 12  0 DC 300u
.ends
*
*modulating signal
*
vmod 1 0 sin(0 4m 10k 0 0 0)
*
*dc
*
vdc 2 0 dc 10m
*
*summer
*
esum 3 0 value {v(1)+v(2)}
*
*carrier signal
*
vcarr 4 0 sin(0 10m 1meg)
*
*first mutiplier
*
emult 5 77 value {v(3)*v(4)}
*
*2nd multiplier
*
xmult 4 5 77 30 40 gilbert_demod
esubtract 6 0 value {v(40)-v(30)}
*
*2nd order butterworth filter to remoove high freuqencies
*parameter definitions for H(s)=a1_/(a11 + a2*s + s*s)
*
.param q=0.707
.param omega=1
.param fc=10k
.param pi=3.14
.param pisquare ={3.14*3.14}
.param a1 ={4*pisquare*omega*omega*fc*fc}
.param a2 ={2*pi*omega*fc/q}
*
*realisation of H(s) with node 7 as output and node 6 as input
*
elow 7 0 LAPLACE {V(6)} = {a1/(a1+a2*s+s*s)}
```

```
*
*transient analysis
*
.tran 5u 800u 600u .1u
.probe
.end
```